Repository: Kotlin/kotlinx.coroutines
File: integration-testing/build.gradle.kts (Kotlin)

Code before:
}

repositories {
    val kotlinRepoUrl = providers.gradleProperty("kotlin_repo_url").orNull
    if (!kotlinRepoUrl.isNullOrBlank()) {
        maven(kotlinRepoUrl)
    }
    mavenLocal()
    mavenCentral()
    maven("https://maven.pkg.jetbrains.space/kotlin/p/kotlin/dev")
    maven("https://redirector.kotlinlang.org/maven/dev")
}

subprojects {
    repositories {
        val kotlinRepoUrl = providers.gradleProperty("kotlin_repo_url").orNull

        if (!kotlinRepoUrl.isNullOrBlank()) {
            maven(kotlinRepoUrl)
        }
        mavenCentral()
        maven("https://maven.pkg.jetbrains.space/kotlin/p/kotlin/dev")
        maven("https://redirector.kotlinlang.org/maven/dev")
        mavenLocal()
    }
}

java {
    sourceCompatibility = JavaVersion.VERSION_1_8
    targetCompatibility = JavaVersion.VERSION_1_8
}

val kotlinVersion = if (extra["build_snapshot_train"] == true) {
    providers.gradleProperty("kotlin_snapshot_version").orNull
        ?: throw IllegalArgumentException("'kotlin_snapshot_version' should be defined when building with snapshot compiler")
} else {
    providers.gradleProperty("kotlin_version").get()
}

val asmVersion = property("asm_version")

dependencies {
    testImplementation("org.jetbrains.kotlin:kotlin-test:$kotlinVersion")
    testImplementation("org.ow2.asm:asm:$asmVersion")
    implementation("org.jetbrains.kotlin:kotlin-stdlib-jdk8")
}

val coroutinesVersion = property("coroutines_version").toString()

sourceSets {
    // An assortment of tests for behavior of the core coroutines module on JVM

Review comment (suggestion):
```suggestion
allprojects {
    repositories {
        val kotlinRepoUrl = providers.gradleProperty("kotlin_repo_url").orNull

        if (!kotlinRepoUrl.isNullOrBlank()) {
            maven(kotlinRepoUrl)
        }
        mavenCentral()
        maven("https://maven.pkg.jetbrains.space/kotlin/p/kotlin/dev")
        maven("https://redirector.kotlinlang.org/maven/dev")
        mavenLocal()
    }
}
```

Code after:
}

allprojects {
    repositories {
        val kotlinRepoUrl = providers.gradleProperty("kotlin_repo_url").orNull
        if (!kotlinRepoUrl.isNullOrBlank()) {
            maven(kotlinRepoUrl)
        }
        mavenCentral()
        maven("https://maven.pkg.jetbrains.space/kotlin/p/kotlin/dev")
        maven("https://redirector.kotlinlang.org/maven/dev")
        mavenLocal()
    }
}

java {
    sourceCompatibility = JavaVersion.VERSION_1_8
    targetCompatibility = JavaVersion.VERSION_1_8
}

val kotlinVersion = if (extra["build_snapshot_train"] == true) {
    providers.gradleProperty("kotlin_snapshot_version").orNull
        ?: throw IllegalArgumentException("'kotlin_snapshot_version' should be defined when building with snapshot compiler")
} else {
    providers.gradleProperty("kotlin_version").get()
}

val asmVersion = property("asm_version")

dependencies {
    testImplementation("org.jetbrains.kotlin:kotlin-test:$kotlinVersion")
    testImplementation("org.ow2.asm:asm:$asmVersion")
    implementation("org.jetbrains.kotlin:kotlin-stdlib-jdk8")
}

val coroutinesVersion = property("coroutines_version").toString()

sourceSets {
    // An assortment of tests for behavior of the core coroutines module on JVM
    create("jvmCoreTest") {
        compileClasspath += sourceSets.test.get().runtimeClasspath
        runtimeClasspath += sourceSets.test.get().runtimeClasspath

        dependencies {
            implementation("org.jetbrains.kotlinx:kotlinx-coroutines-core:$coroutinesVersion")
            implementation("com.google.guava:guava:31.1-jre")
        }
    }

    // Checks correctness of Maven publication (JAR resources) and absence of atomicfu symbols
    create("mavenTest") {